

Utilizzare Inno Setup 1.3.21 con le Applicazioni dB2K

Includes Deploying BDE 5.11

Updated: March, 2001

Ken Mayer, SQA Engineer

dBASE, Inc.

Thanks to Dan Howard, dBVIPS, for Editing ...

Traduzione italiana a cura di Ecosoft Srl – Elia Rivadossi

Suggerimenti, commenti e proposte per migliorare questo documento a elia@ecochim.it

NOTA: Questo HOW TO è rivolto specificatamente alla distribuzione di applicazioni create con dB2K. Se state usando Visual dBASE 7.x esiste un HOW TO diverso che spiega come comportarsi

Contenuti

- [Introduzione](#)
- [Scaricare Inno Setup](#)
- [Installare Inno Setup](#)
- [Partiamo con Inno](#)
 - [Alcune informazioni basilari](#)
 - [Costanti di Inno](#)
 - [Utilizzare altre lingue](#)
- [Creare un'installazione per Vesper 8.10](#)
 - [La sezione "Setup"](#)
 - [La sezione "Dirs"](#)
 - [La sezione "Files"](#)
 - [dB2K Runtime/BDE Installer](#)
 - [La sezione "Run"](#)
 - [La sezione "Icons"](#)
 - [La sezione "Registry"](#)
 - [Opzioni di disinstallazione](#)
 - [Compilare e testare il progetto](#)
- [Installare il BDE con le vostre applicazioni](#)

- [Manipolare le impostazioni del registro](#)
- [Unire i file di configurazione del BDE](#)
- [Collegamenti SQL](#)
- [Creare l'immagine del CD](#)
- [Considerazioni finali](#)
 - [Perchè usare la strada difficile?](#)
 - [Contattare l'autore](#)
 - [Aggiunte di terze parti](#)
- [Lo script completo per Vesper](#)

Introduzione

Ok, avete un'applicazione pronta per essere distribuita, ma non avete InstallShield Express. Cosa fare? Semplicemente – utilizzate la copia di Inno Setup inserita nel Cd di dB2K

Il software è chiamato "Inno Setup", abbreviazione di "Innovative Setup". E' un'applicazione Delphi, che non presenta molteplici interfacce grafiche che vi aiuteranno nel costruire i vostri dischi di distribuzione ma che, è molto semplice da utilizzare, molto veloce e molto flessibile.

C'è una cartella sul vostro CD del dB2K chiamata "Inno" – in questa cartella potete trovare un file .EXE, che è l'immagine del file di installazione di Inno Setup (quando eseguirete questo file installerete la versione corrente disponibile al momento della creazione del CD di dB2K); un'insieme di file per la lingua da utilizzare (*.ISL); e un file di esempio (template.iss) che potete usare per creare i vostri setup personalizzati.

Se non state usando dB2K, o volete ottenere una versione più recente di questo software, visitate il sito internet

<http://www.jordanr.dhs.org/isinfo.htm>

Il sito principale rende disponibile il software, parla delle sue funzionalità, ecc. Se volete potete scaricare direttamente il software seguendo le informazioni sottostanti

[Ritorna all'indice](#)

Scaricare Inno Setup

Ci sono molti siti da dove è possibile scaricare Inno Setup. Io utilizzo questo indirizzo:
<http://www.jordanr.dhs.org/files/site1/is32.exe> (potete cambiare "site1" anche con "site2" o site3")

L'autore, **Jordan Russell**, ha lavorato molto per fornirvi numerose funzionalità. In aggiunta ha anche documentato molto bene il software, con dettagli nell'help on-line

[Ritorna all'indice](#)

Installare Inno Setup

Se avete il dB2K, c'è una cartella sul vostro CD denominata "Inno" – inserite il CD, e se parte l'installazione automatica, usate l'opzione di uscita. A questo punto utilizzando "Gestione Risorse" trovate la cartella "Inno" e fate doppio click sull'eseguibile, seguendo poi le istruzioni qui riportate

Si noti che nella cartella Inno c'è un file .EXE denominato "isetup32-1_3_21.exe" – questo file è la versione disponibile nel momento in cui viene scritto questo documento

Quando avete trovato o scaricato il file , eseguitelo. In questo modo installerete Inno nella cartella di default:

c:\Programmi\Inno Setup 1.3

Quando il software viene aggiornato dall'autore, potete trovare nuove versioni scaricandole dal suo sito web, o in alcuni casi scaricando una semplice patch che aggiorni il software.

[Ritorna all'indice](#)

Partiamo con Inno

Sebbene Inno non abbia un look molto piacevole – non offre le funzionalità di drag and drop che offrono invece altri programmi di installazione – è attualmente abbastanza potente e molto funzionale. Produce inoltre, programmi di installazione *molto veloci*

Alcune informazioni di base

(Molte delle informazioni che seguono sono riprese dal file di help)

Questo programma è semplicemente un editor, che può essere usato per creare script di setup. Lo script creato avrà un'estensione ISS (Inno Setup Script). Il programma è anche un "compilatore" – tenterà di creare il vostro setup basandosi sui contenuti dello script – e se ci sono errori durante il processo di compilazione, vi descriverà il problema riscontrato e il setup non verrà creato.

Lo script è in se un file ASCII, ed è per questo che potete crearlo utilizzando questo programma oppure un qualsiasi programma che vi permetta di creare o modificare script o file di testo.

Il formato dei "comandi" dello script, è simile a quello utilizzato nei files .INI di Windows – se avete familiarità con questi file sarete avvantaggiati.

Molti degli script contengono coppie di nomi/valori – un nome con il rispettivo valore. Questi valori potrebbero essere simili a nomi di file, o cartelle di destinazione per la distribuzione o simili. Questi possono avere anche alcuni parametri – modifiche che volete fare per le vostre distribuzioni.

Notate come Inno Setup realizza un'esecuzione dello script partendo dall'alto – non ci sono istruzioni condizionali (IF/ELSE, ecc), rilevazione di errori in fase di installazione , ecc. Tutto è eseguito in una sequenza specifica, come mostrato in questo documento (e nel file di help di Inno Setup)

Costanti di Inno

Inno supporta l'utilizzo di costanti speciali che rappresentano alcune cartelle sulla macchina dell'utente. Alcune di esse sono definite da voi nella sezione setup.

Notate inoltre come le costanti definite qui sotto non possono essere usate come parametri SOURCE nella sezione "Files"

{app}

La cartella dell'applicazione, che l'utente seleziona nel wizard alla pagina "Selezione della cartella". Per esempio: se usate {app}\MYPROG.EXE in una voce e l'utente ha selezionato C:\MYPROG come cartella dell'applicazione, il Setup tradurrà la stringa in C:\MYPROG\MYPROG.EXE.

{sys}

La cartella di sistema di Windows (System32 in Windows NT). Per esempio: se usate {sys}\CTL3DV2.DLL in una voce e la directory di sistema è C:\WINDOWS\SYSTEM, il Setup tradurrà la stringa in C:\WINDOWS\SYSTEM\CTL3DV2.DLL

{src}

La cartella dove sono collocati i file sorgenti. Per esempio: se usate {src}\MYPROG.EXE in una voce e l'utente sta installando da "S:\", il Setup tradurrà la stringa in "S:\MYPROG.EXE"

{pf}

Files di programma. Il percorso della cartella di sistema dove sono installati i programmi; solitamente C:\Programmi

{cf}

File Comuni. Il percorso della cartella di sistema per i file comuni; solitamente C:\Programmi\File Comuni

Esistono altre costanti che potete trovare documentate nell'help in linea di Inno.

Utilizzare altre lingue

L'autore di Inno Setup, ha scritto il codice in modo che le stringhe contenenti i messaggi usati durante un'installazione siano prese da un file (*.ISL) – questo file contiene le stringhe usate, e può essere cambiato per utilizzare lingue differenti.

Abbiamo inserito nel CD molti di questi files, disponibili al momento della creazione del CD. Potete trovarli nella cartella "Inno" del vostro CD.

Se non trovate la lingua che volete usare tra i files inseriti nel CD di dB2K, potete scaricare nuovi files per lingue specifiche dal sito internet:

<http://www.jordanr.dhs.org/is3rdparty.htm>

Scorrete la pagina fino a quanto non giungete alla sezione lingua. Assicuratevi di essere sul link Inno Setup 1.3 e scaricate il file contenente la lingua che desiderate. E' consigliabile salvare i file scaricati nella stessa cartella in cui è installato Inno Setup o nella cartella della vostra applicazione – io preferisco salvarli nella cartella di Inno Setup anche perché i file di esempio presuppongono che si trovino in questa cartella.

[Ritorna all'indice](#)

Creare un'installazione per Vesper 8.10

Da questo punto in poi, questo documento tratta come creare un'installazione per l'applicazione Vesper 8.10, che è un'applicazione di esempio scritta in dB2K (release 0.1) , disponibile per il download al seguente indirizzo

<http://users.megapathdsl.net/~kmayer/dbase/dbase.htm>

Lo script trattato qui è stato usato per testare alcune opzioni ed ha dato come risultato qualcosa di funzionante – questo significa che questo script di esempio funziona attualmente per la mia applicazione. Ci sono note un po' dappertutto, per spiegarvi quello che ho fatto

Tutto quello che segue è preso dall'help di Inno, tranne le informazioni sull'installazione del BDE e del runtime di dB2K, che sono state prese dal nostro team di sviluppo in dBASE Inc.

Per effettuare dei test di installazione sono andato su una macchina ed ho disinstallato il dB2K, cancellato tutte le cartelle ed i files relativi, rimosso i file di runtime dalla cartella Windows\System e disinstallato il BDE. Ho controllato il registro e mi sono assicurato che non fosse rimasto niente che si riferisse al dB2K, al runtime o al BDE... inoltre dopo ogni prova di installazione ho ripetuto il controllo per essere sicuro di lavorare su una macchina "pulita".

Molti degli argomenti discussi qui sotto sono specifici dello script utilizzato per distribuire l'applicazione Vesper, ma in molte parti si spazia su argomenti generali ...

Potete includere commenti utilizzando il ";" ad inizio riga

La sezione Setup

Come potete immaginare, questa è una sezione molto importante. Guardate i commenti associati ad ogni dichiarazione ...

```
[Setup]
; -----
; Lingua
; Se state utilizzando l'Inglese, ignorate questa parte, altrimenti
; rimuovete il ";" dall'inizio della riga, e
; cambiate il nome del file "default.isl" con il nome
; del file associato alla vostra lingua ...
;MessagesFile=compiler:default.isl
;
; -----
; Per creare dischetti da 3 1/2" settate a 1 il parametro
; attualmente è impostato per creare un CD
DiskSpanning=0
;
; -----
; Definisce la cartella sorgente
; Questa è la cartella sorgente per i files (dove è
; collocata la vostra applicazione?)
SourceDir=C:\Vesper
;
```

```

; -----
; Definisce la cartella di destinazione di default
; Cartella di default, drive corrente:\Programmi
DefaultDirName={pf}\Vesper
;
; -----
; Disabilita l'append sulla cartella
; Se l'utente decide di cambiare cartella, e dopo
; seleziona la stessa cartella proposta sopra, Inno
; aggiungerà il nome della cartella alla fine
; (per esempio: C:\Programmi\Vesper\Vesper
; abilitando questo parametro, la cartella non verrà aggiunta
DisableAppendDir=Yes
;
; -----
; Definisce il nome del Gruppo del menu Programmi che
; verrà creato
; Parametro: nome del gruppo che verrà creato
DefaultGroupName=Vesper 8.10
;
; -----
; Nome dell'applicazione
; Così come apparirà nel programma di Setup (nell'angolo in alto a sinistra)
AppName=Vesper 8.10
;
; -----
; Versione - richiesto
AppVerName=Vesper version 8.10
;
; -----
; Stringa di copyright che verrà visualizzata in basso a destra
; dello schermo di setup
AppCopyright=Copyright © 2000, Golden Stag Productions
;
; -----
; Colore di background (opzionale, per default è impostato a blu)
BackColor=clGreen
; Secondo colore (dissolvenza da uno all'altro - opzionale, per default è
clBlack)
BackColor2=clBlack
; Direzione della dissolvenza (per default dall'alto al basso)
; lefttoright = da sinistra a destra
BackColorDirection=lefttoright
;
; -----
; Icona di disinstallazione (inserita nel gruppo)
AlwaysCreateUninstallIcon=1
;
; -----
; Nome dell'icona di disinstallazione
; Testo associato all'icona di disinstallazione
UninstallIconName=Uninstall Vesper 8.10
;
; -----
; Cartella di output
; La cartella dove verrà creato il programma di installazione
; (questa sarà solitamente sotto la cartella sorgente;
; per esempio C:\Vesper\InstallImage)
OutputDir=InstallImage

```

Ci sono numerose altre opzioni che possono essere configurate , leggete l'help online se siete interessati (esse riguardano ad esempio la possibilità di aggiungere un file di testo contente la licenza d'uso, un file Leggimi che venga aperto prima o dopo l'installazione, e molte altre opzioni ...)

La sezione Dirs

La sezione dedicata alle cartelle è usata per definire le cartelle che devono essere create, oltre alla cartella dell'applicazione... si utilizzano le costanti che risultano definite in tutto lo script. Anche se non è necessario creare esplicitamente le cartelle prima di poter copiarvi dei files dalla sezione [Files], esse non verranno cancellate dall'eventuale disinstallazione se non vengono specificate in questa sezione.

```
[Dirs]
; -----
; Questa sezione è usata per definire tutte le cartelle
; di cui avete bisogno per utilizzare la vostra applicazione
; -----
; Se ad esempio state memorizzando i vostri dati in una cartella
; differente rispetto a quella dell'applicazione (Vesper lo fa)
; potete usare il seguente parametro:
Name: "{app}\tables"
```

La sezione Files

La documentazione definisce questa sezione "opzionale, ma necessaria per molte installazioni". Siccome essa definisce tutti i files che devono essere copiati nel sistema dell'utente, la possiamo considerare indispensabile ...

Ogni file ha una cartella "Sorgente" (Source) e una cartella "Destinazione" (DestDir); ci sono alcuni parametri che dovete esaminare in dettaglio ...

Programma di installazione del runtime di dB2K

NOTA: l'installer del runtime del BDE è memorizzato nel vostro CD del dB2K, in una cartella denominata RUNTIME. Vi raccomando di copiare il file o i files di cui avete bisogno per la vostra applicazione (i) nella directory del dB2K, magari in una cartella nominata Runtime. L'esempio sottostante presuppone che sia stata fatta questa cosa.

Notate inoltre che ci sono 6 programmi di installazione differenti, uno contiene tutte le lingue correntemente supportate, gli altri 5 le lingue specifiche. Se volete distribuire il programma di installazione del runtime/BDE in tutte le lingue usate il file : dB2KruntimeEngine.exe. Se invece volete usare il programma di installazione con lingua specifica (che ha il vantaggio di essere più piccolo) usate uno dei file chiamati dB2KruntimeEngine_XX.exe, dove le lettere XX corrispondono a:

EN = Inglese
DE = Tedesco
ES = Spagnolo
IT = Italiano
JA = Giapponese

Ritorniamo all'esempio:

```
[Files]
; -----
; La sezione files serve per dire a Inno Setup dove:
; a) trovare i files da inserire nel vostro programma di setup
; b) copiare (durante il processo di installazione) i files
; inseriti nel programma di setup
; -----
; ----- dB2K Runtime Files and BDE -----
; il programma di installazione del runtime del dB2K, include il BDE,
; e verrà copiato in una cartella temporanea per essere eseguito in
; seguito ... Vesper usa il runtime in Inglese:
;
Source: "C:\Program Files\dBASE\dB2K 01\Runtime\dB2KRuntimeEngine_EN.exe";
DestDir: "{tmp}"
;
```

```

; -----
; ----- USER DEFINED (per la vostra applicazione) ----
;
; L'applicazione in se, che viene copiata nella "cartella dell'applicazione"
Source: "VESPER.EXE"; DestDir: "{app}"
;
; Non dimenticate il vostro file .INI:
Source: "vesper.ini"; DestDir: "{app}"
;
; Altri files (immagini, ecc)
Source: "Calendar.bmp"; DestDir: "{app}"
Source: "Herald2.ico"; DestDir: "{app}"
Source: "pkzip.exe"; DestDir: "{app}"
Source: "stag.bmp"; DestDir: "{app}"
Source: "vdb7_t.gif"; DestDir: "{app}"
Source: "vdb7ico.gif"; DestDir: "{app}"
Source: "splash.bmp"; DestDir: "{app}"
;
; Il file README; notate il parametro Flags.
Source: "README.TXT"; DestDir: "{app}"; Flags: isreadme
;
; Tabelle, indici e files .dbt copiati nella "cartella applicazione"\tables
; Notate l'utilizzo dell'opzione CopyMode - questa opzione dice all'applicazione
; di copiare le tabelle solo se quelle presenti nel setup sono più recenti di
; quelle derivanti da una precedente installazione
Source: "inventor.dbf"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "inventor.mdx"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "library.dbf"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "library.mdx"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "ranks.dbf"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "ranks.mdx"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "regions.dbf"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "regions.mdx"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "roster.dbf"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "roster.mdx"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "roster.dbt"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "temp.dbf"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "temp.mdx"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "temp.dbt"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "temp2.dbf"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "temp2.dbt"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist

```

La sezione "Run"

Queta sezione è usata per definire quali programmi devono essere eseguiti durante il processo di installazione. Potete eseguire più di un programma e Inno aspetterà, per default, la fine, prima di passare al successivo. Ogni programma definito in questa sezione verrà eseguito nella sequenza assegnata, e dopo l'esecuzione della sezione "Files".

Per i nostri scopi, questa sezione eseguirà l'installazione del BDE e del runtime di dB2K.

Notate che il programma di installazione del runtime potrà porvi una domanda sulla possibilità di memorizzare files nella cartella TEMP – se volete potete eliminare questa domanda (così l'utente finale non la vedrà) aggiungendo l'opzione "-s" nell'istruzione di esecuzione. Nel codice sottoriportato , aggiungeremo il parametro a fine riga:

```

; Parameters: "-s"

```

Così apparirà simile al seguente

```

Filename: "{tmp}\dB2KRuntimeEngine_EN.exe"; Parameters: "-s"

```

Ritorniamo all'esempio:

```

[Run]

```

```

; -----

```



```

; Potete eseguire programmi esterni - saranno eseguiti dopo la
; copia dei files ...
; -----
;           dB2K Runtime/BDE Install
; Questa parte è necessaria per installare il runtime di dB2K e
; il BDE:
; -----
; dB2K release 1 Programma di installazione del BDE/runtime:
;   Prestate attenzione alla lingua, come già fatto notare
;   nella sezione "Files", la lingua può essere importante
;   per la vostra applicazione ...
Filename: "{tmp}\dB2KRuntimeEngine_EN.exe"
;

```

Qualcuno sul newsgroups di Inno Setup ha chiesto la possibilità di visualizzare il nuovo gruppo creato dopo che l'installazione è stata completata. Questa è una soluzione elegante, specialmente fino a quando l'installazione risulta essere molto veloce. Potete fare ciò aggiungendo le seguenti linee:

```

; Visualizza il gruppo quando creato:
Filename: "{group}\"; Flags: shellexec

```

La sezione "Icons"

Come la sezione files, anche questa può non essere indispensabile, ma è molto vicino ad essere più che necessaria!

```

[Icons]
; -----
; Quali icone usare nel gruppo di programma
; -----
; Icona applicazione
Name: "{group}\Vesper 8.10"; Filename: "{app}\Vesper.exe"; WorkingDir: "{app}"
; -----
; BDE Administrator - se volete far conoscere agli utenti
; che esiste ...
; Notate che il problema è che qui si assume che l'utente
; non installi il BDE in una cartella diversa da quella
; proposta - su questo non avete il pieno controllo ...
Name: "{group}\BDE Administrator"; Filename: "{cf}\Borland
Shared\BDE\BDEAdmin.exe"; WorkingDir: "{cf}\Borland Shared\BDE"
; -----
; Readme:
Name: "{group}\Readme.txt"; Filename: "{app}\readme.txt"; WorkingDir: "{app}"

```

Se volete potete aggiungere altre icone per altri programmi.

La sezione "Registry"

Qui è dove potete inserire tutte le eventuali chiavi di registro di cui la vostra applicazione bisogna. Se non ne state usando, potete escludere questa sezione dal vostro script di Inno Setup.

Opzioni di disinstallazione

Il programma di disinstallazione creato da Inno Setup è memorizzato nella stessa cartella della vostra applicazione e si chiama: UNINS000.EXE, e comprende anche altri numerosi file (UNINS000.DAT e UNINS000.MSG). La disinstallazione è veloce. Si pone l'attenzione su come questo programma non disinstalli il BDE e il runtime di dB2K, che sono installati tramite l'utilizzo di un altro programma...

Compilazione e test

Fate click sul menu "File" e prima di tutto salvate il vostro script (sempre una buona idea!). A questo punto cliccate ancora sul menu "File" e selezionate la voce "Compile" (o premete F9). Per default, questo programma crea le immagini in una cartella denominata "Output" (come sottodirectory della cartella in cui si trova il vostro script), ma l'esempio che stiamo seguendo attualmente, dice di creare la cartella "InstallImage"

come sottodirectory della vostra cartella applicazione (C:\VESPER). La cartella di output contenente l'immagine dei dischi di installazione sarà quindi C:\VESPER\INSTALLIMAGE

Se avete impostato l'utilizzo di un CD (singola immagine del file di setup), avrete solo il file SETUP.EXE ...

Se il parametro DiskSpan è stato impostato a 1 (floppy disk), avrete in questo caso 14 immagini per dischetti da 3 1/2" con questi nomi:

Setup.1
Setup.2
ecc, ecc

Per eseguire un'installazione da floppy disk copiate il file SETUP.EXE e il file SETUP.1 sul primo disco e, le immagini degli altri dischi su altri dischetti. Se utilizzate un'installazione del runtime di dB2K e del BDE con tutte le lingue serviranno 14 dischi.

La ragione per cui (se usate l'installazione di tutte le lingue) ci sono 14 immagini per l'installazione tramite dischetti è che il runtime e il BDE sono configurati per installare 6 lingue differenti – all'utente viene richiesto, durante l'installazione del runtime, quale lingua installare. Questo per garantire che siano installate le versioni corrette dei file di linguaggio del runtime, del BDE, dei files di help, ecc.

La ragione per cui ci sono i file di installazione del runtime specifici per ogni lingua è quella di ridurre la dimensione delle vostre installazioni (questo diventa utile, anche e soprattutto, per le distribuzioni tramite internet)

[Ritorna all'indice](#)

Installare il BDE con le vostre applicazioni

Manipolare i settings del registro

Per quanto riguarda le impostazioni del registro, dovrete innanzitutto considerare se avete veramente bisogno di manipolarle. Un oggetto che è veramente importante è il driver che definisce la lingua di default. Se non impostate questo parametro, il driver dBASE di default sarà "ascii ANSI". L'applicazione Vesper usa "Western European ANSI" ("Weurope' Ansi), in quanto alcuni caratteri usati richiedono proprio questo driver ... Per salvare le impostazioni di registro del BDE utilizzati dalla vostra applicazione, potete usare un file .REG (creato dal "REGEDIT" – l'editor di registro del vostro computer) e copiare le chiavi necessarie. **Nota importante:** Gli aliases del BDE non sono memorizzati nel registro – guardate le note successive in questo documento per avere informazioni a riguardo.

NOTA: se state usando Windows 2000 dovete usare REGEDIT4, in quanto con il REGEDIT standard, sembra che le chiavi di registro vengano memorizzate in un formato differente, e se state creando un'installazione, diciamo, per una macchina con Win 98, le chiavi di registro non verranno riconosciute.

Un file .REG è semplicemente un file di testo, e questo vi permette quindi di aprirlo ed editarlo direttamente prima di distribuirlo. Quello che segue è un esempio di quello che potete fare. Potreste, ad esempio, assicurarvi che il driver dBASE di lingua di default sia quello che avete preferito usare nella vostra applicazione. Questo può essere fatto ricercando la seguente chiave di registro.

- Eseguite REGEDIT.EXE da Windows (cliccate sul bottone "Start", selezionate "Esegui" , digitate "REGEDIT", e premete <Enter> o cliccate sul bottone "OK")
- Nella vista sulla sinistra, cliccate sul "+" di: HKEY_LOCAL_MACHINE
- In questo modo verranno visualizzate maggiori informazioni, fate click sul "+" di. Software
- Click sul "+" di. Borland
- Click sul '+' di Database Engine
- Click sul "+" di: Settings

- Click sul “+” di: DRIVERS
- Click sul “+” di: DBASE
- Utilizzando il menu “Registro di configurazione” e la voce “Esporta file del Registro di Configurazione” – assegnate un percorso, e quindi un nome al file (BDEStuff”) - creerete un piccolo file .REG con alcune impostazioni, incluse le informazioni sul driver di lingua.

Se aprite il file con un editor di testo, (notepad, wordpad, o l'editor sorgenti di dBASE) dovrebbe apparire così:

```
REGEDIT4
```

```
[HKEY_LOCAL_MACHINE\Software\Borland\Database Engine\Settings\DRIVERS\DBASE]
```

```
[HKEY_LOCAL_MACHINE\Software\Borland\Database Engine\Settings\DRIVERS\DBASE\INIT]
"VERSION"="1.0"
"TYPE"="FILE"
"LANGDRIVER"="DBWINWE0"
```

```
[HKEY_LOCAL_MACHINE\Software\Borland\Database Engine\Settings\DRIVERS\DBASE\TABLE CREATE]
"LEVEL"="7"
"MDX BLOCK SIZE"="1024"
"MEMO FILE BLOCK SIZE"="1024"
```

Se volete, potete editare tutte queste linee. Vorrei però raccomandarvi di lasciarle “così come sono”. La chiave “LANGDRIVER” sarà impostata in base alle impostazioni correnti inserite nel BDE Administrator per le tabelle dBASE.

Potete anche esportare la sezione HKEY_LOCAL_MACHINE\Software\Borland\Database Engine\Settings\SYSTEM\INIT. Questa sezione contiene le variabili del BDE e le impostazioni dei file come MaxFileHandles, MaxBufferSize etc. Questo potrebbe essere importante se utilizzate impostazioni più elevate rispetto ai valori impostati di default dal BDE

Notate come ogni esportazione del registro debba essere inserita in un file separato. Non inserite esportazioni diverse in un singolo file .REG in quanto i settaggi potrebbero non essere inseriti correttamente nel registro.

Per aggiornare il registro della macchina dell'utente, copiate questi files nella cartella contenente i files sorgente. Poi, nella sezione “FILES” dello script di Inno Setup che state creando, aggiungete quanto segue:

```
; -----
; Impostazioni del registro per il BDE (si assume che il file .REG
; sia nella stessa cartella della vostra applicazione:
Source: "BDEStuff.reg"; DestDir: "{tmp}"; Flags: deleteafterinstall
```

Questo copierà il file nella cartella temp del computer dell'utente. Poi, nella sezione “RUN”, aggiungete quanto segue, dopo l'installazione del Runtime di dB2K (se inserito prima dell'installazione del runtime, le nuove chiavi verrebbero sovrascritte ...):

```
; -----
; "Esegue" il file di registro con i settings del BDE...
Filename "{win}\REGEDIT.EXE"; Parameters "/S ""{tmp}\BDEStuff.reg""
```

Questa istruzione farà sì che il file .REG venga “eseguito” da Windows *dopo* che il BDE è stato installato. In questo modo si risolvono i problemi dovuti al settaggio del BDE prima di installare il BDE stesso (che causa la sovrascrittura dei settaggi del registro). Notate come il programma REGEDIT venga eseguito con l'opzione /S (“silenziosa”), in modo che non appaiano finestre di dialogo (*Si ringrazia per questa informazione Jordan Russel*)

Unire i file di configurazione del BDE

Jordan Russel (autore di Inno Setup) non è sicuro su come procedere con gli aliases del BDE, con i file di configurazione, ecc, ecc. Vic McClung, uno sviluppatore che utilizza dBASE per molte cose, ha impiegato un po' di tempo per capire come unire i file di configurazione del BDE utilizzando BDE API (Application Programming Interface), ed ha creato un semplice programma dBASE che è fornito insieme a questo documento. Il programma si chiama "Merge.prg" (disponibile nella cartella Inno del vostro CD di dB2K)

Per usare questo programma, ci sono numerosi passi da eseguire; essi sono riportati qui sotto.

1. **Create un file .CFG personalizzato**

I file di configurazione personalizzati sono creati utilizzando il BDE Administrator. Assicuratevi che nella vostra configurazione corrente del BDE siano inseriti gli aliases di cui avete bisogno per la vostra applicazione

Usate il menu "Object" e selezionate la voce "Save as Configuration...". Questa opzione di menu vi chiederà il nome del file e la posizione per la vostra nuova configurazione. Potreste inserire la vostra configurazione personalizzata nella cartella dell'applicazione (questo renderà più semplice la distribuzione). Dategli un nome che abbia senso per voi, come ad esempio "MiaApplicazione.cfg" (dove "MiaApplicazione" è il nome della vostra applicazione)

Adesso, potrete notare come il file .CFG attualmente in uso dal BDE Administrator è quello personalizzato. Passate alla lista degli aliases nello schermo principale (cliccate su "Databases" e poi sul "+" della parola "Databases" per vedere la lista completa degli aliases).

Dovete cancellare gli aliases che non sono importanti nella vostra applicazione – questa operazione non avrà alcun effetto sulla configurazione originale – la ristabiliremo in un baleno, Potete cancellare questi aliases cliccando su di essi e premendo CTRL+D o usate il menu "Object" e la voce "Delete"

Se avete un alias che non si cancella, molto probabilmente è un impostazione di un driver ODBC – non potete cancellarlo, ma se il vostro utente avrà installato lo stesso driver ODBC non vedrà queste voci – in altre parole, non dovete preoccuparvi di questo problema.

A questo punto avete il vostro file di configurazione e dovete quindi reimpostare il BDE Administrator usando il file originale "IDAPI.CFG" dal quale siete partiti. Selezionate il menu "Object" e la voce "Open Configuration ...". Selezionate il percorso dove è installato il BDE – per default

```
C:\Program Files\Common Files\Borland\BDE
```

e selezionate "Idapi.cfg", poi cliccate su "open" – vedrete la lista dei databases cancellarsi – se in seguito riaprirete il BDE ritroverete tutto gli aliases che avevate prima di fare queste operazione

2. **Compilare il programma "Merge"**

E' fatto molto semplicemente - lanciate dB2K, trovate la cartella dove avete il file Merge.prg usando il Navigatore e nella finestra Comandi digitate quanto segue:

```
compile _dbwinhome+"samples\registry.prg"  
compile merge  
build merge.pro, _dbwinhome+"samples\registry.pro" to merge
```

A questo punto avete il file MERGE.EXE, che è l'applicazione che prende il vostro file .CFG e lo unisce ai file .CFG del BDE quando installate l'applicazione

3. **Modificare lo script di Inno Setup per la Vostra applicazione**

Ci sono alcune modifiche da effettuare allo script

A. La sezione "Files"

Questa è la sezione che usate per definire la distribuzione dei files. Aggiungete quanto segue:

```
    ; percorso del file MERGE.EXE che avete appena creato  
  
    Source: "\PathToExe\MERGE.EXE"; DestDir: "{tmp}"  
    ; percorso del file personalizzato .CFG, che avete appena  
creato
```

```
; se non specificate il percorso, Inno lo cercherà  
; nella cartella applicazione, dove sono i files .EXE  
; della vostra applicazione e dove sono gli altri files.  
Source: "\PathToCFG\MyApp.CFG"; DestDir: "{tmp}"
```

B. La sezione "Run"

Qui è dove dite ad Inno Setup di lanciare gli eseguibili di cui avete bisogno. Aggiungete quanto segue dopo la parte dedicata al runtime, altrimenti il file MERGE.EXE non verrà eseguito.

```
Filename: "{tmp}\Merge.exe"; Parameters: "MyApp.CFG"
```

Quando avrete finito, salvate il vostro script di Inno Setup e sarete pronti per testare la distribuzione.

Un altro metodo

Se non siete soddisfatti delle funzionalità di questo programma, della creazione di un file di configurazione personalizzato, ecc, esiste una classe custom nella dBASE Users'Function Library (nella Knowledgebase di dB2K) chiamata "BDEAlias.cc", che può essere usata per creare alias BDE dalla vostra applicazione, quando viene eseguita. Funziona molto bene e vi fornisce un altro metodo per gestire gli aliases del BDE.

[Ritorna all'indice](#)

SQL Links Drivers

Quando l'utente installa dB2K e il runtime usando i programmi forniti (standard) , tutti i file SQL Links necessari verranno installati sulla macchina. Questo significa che i driver per Oracle, InterBase, Sybase, e altri SQL verranno installati – non dovete fare niente di particolare per installarli

Questo non significa che, poiché i driver SQL sono installati, il vostro utente abbia tutto ciò che serve per un server SQL specifico (per esempio Oracle, ecc). SQL Links sono i drivers nativi per questi servers, ma non permettono alle vostre applicazioni di lavorare senza che il software client/server appropriato sia installato. Consultate la documentazione allegata al vostro software SQL per avere maggiori dettagli.

[Ritorna all'indice](#)

Creare l'immagine del CD

Quando eseguite il vostro software per la masterizzazione, qualunque esso sia, dovete impostare quali file inserire nel CD (file immagine), includendo eventualmente altri software, se pensate che siano necessari (e avete spazio a sufficienza sul CD).

Una cosa che non ho trovato documentata nel mio software di masterizzazione era, come dire al sistema operativo quale programma eseguire quando l'opzione AUTORUN di Windows è attiva (impostazione di default).

La questione è molto semplice. C'è un file su ogni CD che è usato quando l'opzione "autorun" è attiva, chiamato AUTORUN.INF. Questo file deve contenere queste 3 linee – modificatele per la vostra applicazione – il file deve essere creato/salvato come un file di testo DOS:

```
[autorun]  
OPEN=setup.exe  
ICON=MyApp.ico
```

[Ritorna all'indice](#)

Considerazioni finali

La distribuzione di Vesper è indirizzata per utenti singoli, un'applicazione relativamente semplice. Ci sono opzioni nel programma di setup che dovrete esaminare per installare applicazioni più complesse ... lo scopo di questo documento era solo quello di darvi le informazioni di base per partire...

Perché usare la strada difficile?

Beh, perché è stato un modo per imparare come funziona qualcosa... anche se, a lungo andare, utilizzare un programma con interfaccia grafica può risultare più semplice. ☺

Date un'occhiata alla sezione "terze parti" qui sotto ...

Contattare l'autore

Per assistenza su Inno Setup, contattate per favore, l'autore utilizzando questo indirizzo :

<http://www.jordanr.dhs.org/isinfo.htm>

In aggiunta, potete utilizzare il newsgroups che fornisce supporto per Inno Setup:

<http://www.jordanr.dhs.org/newsgroups.htm>

Per informazioni su dB2K – se avete comprato dBASE, consultate il CD che contiene tutta la Knowledgebase, oppure visitate il sito www.dbase.com e cercate la sezione Knowledgebase, che sicuramente sarà più aggiornata rispetto alla versione inserita sul CD.

Add Ons di terze parti

TaFWeb Software ha creato un'interfaccia grafica (GUI) per Inno chiamata "ScriptMaker". Può essere trovata al seguente indirizzo :

<http://www.tafweb.com/scriptmaker.html>

Esistono inoltre numerose altre interfacce grafiche; visitate la pagina <http://www.jrsoftware.org/is3rdparty.htm> per maggiori informazioni.

[Ritorna all'indice](#)

Lo script completo di Vesper 8.10

Lo script funziona con l'applicazione Vesper 8.10. Comunque, nella cartella di Inno sul vostro CD, sono inseriti anche altri script più generici. ("template.iss")

```
; This is a test Inno Setup Script for the Vesper 8.10
; Application to see if we can get this to work
; and install everything as needed
;
; Note: This script contains installation instructions
; to handle deploying the runtime for dB2K release 1.
;
[Setup]
; To Create 3 1/2" disks set this to 1, it is set currently
; to create a a CD image:
DiskSpanning=0

; Source folder for files (where is your application?):
SourceDir=C:\Vesper8

; Default directory name, under the drive:\Program Files
; location
DefaultDirName={pf}\Vesper

; If the user decides to change directories, and then
; selects the same directory as above, Inno will append
; the directory name to the end
; (i.e., C:\Program Files\Vesper\Vesper )
```

```

; By disabling this, it will not append the directory ...
DisableAppendDir=Yes

; Name of group that is created
DefaultGroupName=Vesper 8.10
;
; Items involved in install program display:
; Application Name -- as it will appear in Setup program (upper left corner)
AppName=Vesper 8.10
; Version -- required
AppVerName=Vesper version 8.10
; Copyright for display in lower right of setup screen
AppCopyright=Copyright © 2000, Golden Stag Productions
; Background color (optional, default is clBlue)
BackColor=clGreen
; Second color (fade from one to other -- optional, default is clBlack)
BackColor2=clBlack
; Direction to fade (default is toptobottom)
BackColorDirection=lefttoright
;
; Uninstall icon (placed in the group):
AlwaysCreateUninstallIcon=1
; Uninstall Icon Name:
UninstallIconName=Uninstall Vesper 8.10
; Output directory -- where the install program
; gets created (this will be "under" the source, i.e.,
; C:\Vesper\InstallImage):
OutputDir=InstallImage

[Dirs]
; IF you are storing the data in a different folder than the folder
; the application is installed to (Vesper does):
Name: "{app}\tables"

[Files]
; ----- dB2K Runtime Files and BDE -----
; dB2K runtime installer, includes the BDE, and will be installed
; to the temp folder, and run later ...
Source: "C:\Program Files\dBASE\dB2K 01\Runtime\dB2KRuntimeEngine_EN.exe";
DestDir: "{tmp}"

; The application itself, to the application folder:
Source: "VESPER.EXE"; DestDir: "{app}"
; Don't forget your .INI file:
Source: "vesper.ini"; DestDir: "{app}"
;
; Misc. files
Source: "Calendar.bmp"; DestDir: "{app}"
Source: "Herald2.ico"; DestDir: "{app}"
Source: "pkzip.exe"; DestDir: "{app}"
Source: "stag.bmp"; DestDir: "{app}"
Source: "vdb7_t.gif"; DestDir: "{app}"
Source: "vdb7ico.gif"; DestDir: "{app}"
Source: "splash.bmp"; DestDir: "{app}"
;
; Readme file, note the flags parameter
Source: "README.TXT"; DestDir: "{app}"; Flags: isreadme
;
; Tables, indexes and .dbts to the application\tables folder:
Source: "tables\inventor.dbf"; DestDir: "{app}\tables"; CopyMode:
onlyifdoesntexist
Source: "tables\inventor.mdx"; DestDir: "{app}\tables"; CopyMode:
onlyifdoesntexist

```

```

Source: "tables\library.dbf"; DestDir: "{app}\tables"; CopyMode:
onlyifdoesntexist
Source: "tables\library.mdx"; DestDir: "{app}\tables"; CopyMode:
onlyifdoesntexist
Source: "tables\ranks.dbf"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "tables\ranks.mdx"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "tables\regions.dbf"; DestDir: "{app}\tables"; CopyMode:
onlyifdoesntexist
Source: "tables\regions.mdx"; DestDir: "{app}\tables"; CopyMode:
onlyifdoesntexist
Source: "tables\roster.dbf"; DestDir: "{app}\tables"; CopyMode:
onlyifdoesntexist
Source: "tables\roster.mdx"; DestDir: "{app}\tables"; CopyMode:
onlyifdoesntexist
Source: "tables\roster.dbt"; DestDir: "{app}\tables"; CopyMode:
onlyifdoesntexist
Source: "tables\temp.dbf"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "tables\temp.mdx"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "tables\temp.dbt"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "tables\temp2.dbf"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
Source: "tables\temp2.dbt"; DestDir: "{app}\tables"; CopyMode: onlyifdoesntexist
;
; -----
; BDE Registry settings
Source: "BDEStuff.reg"; DestDir: "{tmp}"

[Run]
; -----
; dB2K release 1 BDE/Runtime Installer:
Filename: "{tmp}\dB2KRuntimeEngine_EN.exe"
;
; -----
; Registry file ...
Filename: "{win}\REGEDIT.EXE"; Parameters: "/S ""{tmp}\BDEStuff.reg""
;
; -----
; Display the group when done:
Filename: "{group}\"; Flags: shellexec

[Icons]
; Application icon
Name: "{group}\Vesper 8.10"; Filename: "{app}\Vesper.exe"; WorkingDir: "{app}"
; BDE Administrator - if you wish to allow your users to even
; know this exists ...
; Note that the problem here is that this assumes the user
; didn't install the BDE to some other location -- you
; have no real control over this ...
Name: "{group}\BDE Administrator"; Filename: "{cf}\Borland
Shared\BDE\BDEAdmin.exe"; WorkingDir: "{cf}\Borland Shared\BDE"
; Readme:
Name: "{group}\Readme.txt"; Filename: "{app}\readme.txt"; WorkingDir: "{app}"

[Registry]
;
; None is necessary if using the dB2K runtime installer, unless
; your application needs a registry setting ...

; -----
; End of Vesper.iss Inno Setup script
; -----

```

[Ritorna all'indice](#)